Soluzioni del compito di Metodi Matematici e Calcolo per Ingegneria del 28 giugno 2005

Enrico Bertolazzi

Trasformata di Laplace

```
> restart:
  with(inttrans) :
```

Data la seguente funzione

```
> f := t -> 1+exp(-t) ;
```

$$f := t \to 1 + \mathbf{e}^{(-t)}$$

Usando le regole di trasformazione calcolare le trasformate delle funzioni

```
> f1 := t -> f(3*t) ;
  f2 := t -> f(2*t)*exp(-t) ;
  f3 := t -> D(D(f))(t) ;
```

$$f1 := t \to f(3\ t)$$

$$f2 := t \to f(2\ t)\ \mathbf{e}^{(-t)}$$

$$f3 := t \to \mathrm{D}(\mathrm{D}(f))(t)$$

Trasformate con le primitive Maple

```
> laplace(f(t),t,s);
  laplace(f1(t),t,s);
  laplace(f2(t),t,s);
  laplace(f3(t),t,s);
```

$$\frac{\frac{1}{s} + 2}{1 + s}$$

$$\frac{2\ s + 3}{s\ (s + 3)}$$

$$\frac{2\ (s + 2)}{(1 + s)\ (s + 3)}$$

$$\frac{1}{1 + s}$$

Soluzione di ODE con Laplace

```
> restart:
  with(inttrans) :
```

Data la seguente equazione differenziale

```
> ode := diff(y(x),x,x,x)-2*diff(y(x),x)=exp(-x) ;
```

$$ode := \left(\frac{d^3}{dx^3}y(x)\right) - 2\left(\frac{d}{dx}y(x)\right) = \mathbf{e}^{(-x)}$$

Con dato iniziale

```
> y0, yp0, ypp0 := 1,2,3 ;
```

$$y0, yp0, ypp0 := 1, 2, 3$$

Calcolare le soluzioni con le trasformate di Laplace.

Trasformo la equazione differenziale con la trasformata di Laplace

```
> sode := laplace(ode,x,s) ;
```

$$sode := s^3\ laplace(y(x), x, s) - \mathrm{D}^{(2)}(y)(0) - s\,\mathrm{D}(y)(0) - s^2\,y(0) - 2\,s\,laplace(y(x), x, s) + 2\,y(0) = \frac{1}{1+s}$$

Risolvo la equazione per y(s)

```
> lode := isolate(sode,laplace(y(x),x,s));
```

$$lode := laplace(y(x), x, s) = \frac{\frac{1}{1+s} + \mathrm{D}^{(2)}(y)(0) + s\,\mathrm{D}(y)(0) + s^2\,y(0) - 2\,y(0)}{s^3 - 2\,s}$$

Applico le condizioni iniziali ottenendo y(s)

```
> ly := subs( y(0)=y0,D(y)(0)=yp0,D(D(y))(0)=ypp0,rhs(lode)) ;
```

$$ly := \frac{\frac{1}{1+s} + 1 + 2\,s + s^2}{s^3 - 2\,s}$$

Espansione in fratti semplici

```
> convert(ly, parfrac, s);
```

$$\frac{s+3}{s^2-2} - \frac{1}{s} + \frac{1}{1+s}$$

Antitrasformo per ottenere la equazione y(x)

```
> res := invlaplace(ly,s,t) ;
```

$$res := -1 + \cosh(\sqrt{2}\ t) + \frac{3}{2}\sqrt{2}\ \sinh(\sqrt{2}\ t) + \mathbf{e}^{(-t)}$$

Soluzione di un sistema di ODE con Laplace

```
> restart:
  with(inttrans) :
```

Dato il seguente sistema di equazioni differenziali

```
> ode1 := diff(y(x),x)-diff(z(x),x)=cos(x) ;
  ode2 := diff(y(x),x)+2*diff(z(x),x)=0 ;
```

$$ode1 := \left(\frac{d}{dx}\,y(x)\right) - \left(\frac{d}{dx}\,z(x)\right) = \cos(x)$$

$$ode2 := \left(\frac{d}{dx}\,y(x)\right) + 2\left(\frac{d}{dx}\,z(x)\right) = 0$$

Con dato iniziale

```
> y0, z0 := 1,2 ;
```

$$y0, z0 := 1, 2$$

Calcolare le soluzioni con le trasformate di Laplace.

Trasformo le equazioni differenziale con la trasformata di Laplace

```
> sode1 := laplace(ode1,x,s) ;
  sode2 := laplace(ode2,x,s) ;
```

$$sode1 := s\, laplace(y(x), x, s) - y(0) - s\, laplace(z(x), x, s) + z(0) = \frac{s}{s^2+1}$$

$$sode2 := s\, laplace(y(x), x, s) - y(0) + 2\, s\, laplace(z(x), x, s) - 2\, z(0) = 0$$

Risolvo la equazione per y(s), z(s)

```
> RES := solve({sode1,sode2},{laplace(y(x),x,s),laplace(z(x),x,s)});
```

$$RES := \left\{ laplace(y(x), x, s) = \frac{2\,s + 3\,y(0)\,s^2 + 3\,y(0)}{3\,s\,(s^2+1)},\; laplace(z(x), x, s) = \frac{-s + 3\,z(0)\,s^2 + 3\,z(0)}{3\,s\,(s^2+1)} \right\}$$

Applico le condizioni iniziali ottenendo y(s), z(s)

```
> SOL := subs(RES,y(0)=y0,z(0)=z0,<laplace(y(x),x,s),laplace(z(x),x,s)>);
```

$$SOL := \begin{bmatrix} \dfrac{2\,s + 3\,s^2 + 3}{3\,s\,(s^2+1)} \\[2ex] \dfrac{-s + 6\,s^2 + 6}{3\,s\,(s^2+1)} \end{bmatrix}$$

Antitrasformo per ottenere y(x), z(x)

```
> yy := invlaplace(SOL[1],s,x) ;
  zz := invlaplace(SOL[2],s,x) ;
```

$$yy := \frac{2}{3}\sin(x) + 1$$

$$zz := -\frac{1}{3}\sin(x) + 2$$

Espansione in fratti semplici per controllo

```
> convert(SOL[1], parfrac, s);
  convert(SOL[2], parfrac, s);
```

$$\frac{2}{3(s^2+1)} + \frac{1}{s}$$

$$-\frac{1}{3(s^2+1)} + \frac{2}{s}$$

Soluzione di ricorrenza con trasformata zeta

```
> restart:
```

```
> expand((x+1)*(x-2));
```

$$x^2 - x - 2$$

Risolvere la seguente ricorrenza

```
> RIC := f(n+2) = f(n+1) + 2 * f(n) + 2  ;
```

$$RIC := f(n+2) = f(n+1) + 2\,f(n) + 2$$

Con dato iniziale

```
> INI := f(0)=0,f(1)=1 ;
```

$$INI := f(0) = 0, f(1) = 1$$

Usando le primitive si maple:

```
> rsolve({RIC,INI}, f(k));
```

$$2^k - 1$$

```
> simplify(%);
```

$$2^k - 1$$

Usando la Z-trasformata (occhio negli appunti abbiamo Z(f) = sum(f(n)*z^n mentre maple usa Z(f) = sum(f(n)*w^{-n}
Quindi per confrontare le trasformate bisogna trasfromare w -> 1/z;

```
> zRIC := ztrans(RIC,n,w):
  simplify(subs( ztrans(f(n),n,w)=f(z), w=1/z, zRIC)) ;
```

$$-\frac{-f(z) + f(0) + f(1)\,z}{z^2} = \frac{-f(z) - f(z)\,z + f(0) - f(0)\,z + 2\,f(z)\,z^2 - 2\,z}{z\,(-1+z)}$$

Ricavo f(w) [f(1/z)]

```
> zRICrhs := isolate(zRIC,ztrans(f(n),n,w)):
  simplify(subs( ztrans(f(n),n,w)=f(z), w=1/z, zRICrhs)) ;
```

$$f(z) = \frac{f(0) - 2\, f(0)\, z + f(1)\, z - f(1)\, z^2 + f(0)\, z^2 + 2\, z^2}{(-1 + z)\, (-1 + z + 2\, z^2)}$$

Applico le condizioni iniziali

```
> zRICrhsINI := subs(INI,zRICrhs):
  simplify(subs( ztrans(f(n),n,w)=f(z), w=1/z, zRICrhsINI)) ;
```

$$f(z) = \frac{z}{(2\, z - 1)\, (-1 + z)}$$

Conversione in fratti semplici

```
> convert(%, parfrac);
```

$$f(z) = \frac{1}{-1 + z} - \frac{1}{2\, z - 1}$$

Inversione della Z-trasformata

```
> invztrans(zRICrhsINI,w,k) ;
```

$$f(k) = 2^k - 1$$

Soluzione di un sistema non lineare con Newton

```
> restart:
  with(VectorCalculus):
Warning, the assigned names `<,>` and `<|>` now have a global binding

Warning, these protected names have been redefined and unprotected: `*`,
`+`, `.`, D, Vector, diff, int, limit, series
```

Sistema non lineare

```
> f := (x+y)^2-1 ;
  g := (x-y)^2-1 ;
```

$$f := (x + y)^2 - 1$$

$$g := (x - y)^2 - 1$$

Soluzione esatta

```
> solve({f,g},{x,y}) ;
```

$$\{x = 1, y = 0\}, \{x = 0, y = -1\}, \{x = -1, y = 0\}, \{y = 1, x = 0\}$$

Matrice Jacobiano

```
> J := Jacobian([f,g],[x,y]) ;
```

$$J := \begin{bmatrix} 2x+2y & 2x+2y \\ 2x-2y & -2x+2y \end{bmatrix}$$

Schema di Newton

```
> Newton_update := <x,y>-J^(-1).<f,g> ;
```

$$Newton\_update := \left(x - \frac{(x+y)^2-1}{4(x+y)} + \frac{(x-y)^2-1}{4(-x+y)} \right) e_x + \left(y - \frac{(x+y)^2-1}{4(x+y)} - \frac{(x-y)^2-1}{4(-x+y)} \right) e_y$$

Schema di Newton per questo sistema non lineare

```
> x[k+1]=simplify(subs(x=x[k],y=y[k],Newton_update[1])) ;
  y[k+1]=simplify(subs(x=x[k],y=y[k],Newton_update[2])) ;
```

$$x_{k+1} = \frac{x_k (x_k^2 - y_k^2 + 1)}{2(x_k^2 - y_k^2)}$$

$$y_{k+1} = \frac{y_k (x_k^2 - y_k^2 - 1)}{2(x_k^2 - y_k^2)}$$

Tre iterate a partire da (1,2)

```
> x[0],y[0]:= 1,2 ;
```

$$x_0, y_0 := 1, 2$$

Prima iterata

```
> x[1] := evalf(subs(x=x[0],y=y[0],Newton_update[1])) ;
  y[1] := evalf(subs(x=x[0],y=y[0],Newton_update[2])) ;
```

$$x_1 := 0.3333333333$$

$$y_1 := 1.333333333$$

Seconda iterata

```
> x[2] := evalf(subs(x=x[1],y=y[1],Newton_update[1])) ;
  y[2] := evalf(subs(x=x[1],y=y[1],Newton_update[2])) ;
```

$$x_2 := 0.06666666665$$

$$y_2 := 1.066666666$$

Terza iterata

```
> x[3] := evalf(subs(x=x[2],y=y[2],Newton_update[1])) ;
  y[3] := evalf(subs(x=x[2],y=y[2],Newton_update[2])) ;
```

$$x_3 := 0.003921568400$$

$$y_3 := 1.003921568$$

Problema di Minimo Vincolato

```
> restart:
  with(LinearAlgebra):
  with(Optimization):
  with(VectorCalculus):
```

Warning, the names `&x`, CrossProduct and DotProduct have been rebound

Warning, the assigned names `<,>` and `<|>` now have a global binding

Warning, these protected names have been redefined and unprotected: `*`, `+`, `.`, D, Vector, diff, int, limit, series

Minimizzare la seguente funzione

```
> f := x+y+z^2;
```

$$f := x + y + z^2$$

Soggetta ai seguenti vincoli

```
> v := [x+y+z=3,x-2*z^2=0] ;
```

$$v := [x + y + z = 3, x - 2\, z^2 = 0]$$

Soluzione con le primitive Maple

```
> Minimize(f, v );
```

$$[2.75000000000000044, [x = 0.499999999280939022, z = 0.499999999641270842, y = 2.00000000107779030]]$$

Uso dei moltiplicatori di Lagrange

```
> v1 := lhs(v[1])-rhs(v[1]) ;
  v2 := lhs(v[2])-rhs(v[2]) ;
```

$$v1 := x + y + z - 3$$

$$v2 := x - 2\, z^2$$

```
> g := f - lambda*v1 - mu*v2 ;
```

$$g := x + y + z^2 - \lambda\, (x + y + z - 3) - \mu\, (x - 2\, z^2)$$

Sistema non lineare da risolvere

```
> F := Gradient(g,[x,y,z,lambda,mu]) ;
```

$$F := (1 - \lambda - \mu)\, \overline{e}_x + (1 - \lambda)\, \overline{e}_y + (2\, z - \lambda + 4\, \mu\, z)\, \overline{e}_z + (-x - y - z + 3)\, \overline{e}_\lambda + (-x + 2\, z^2)\, \overline{e}_\mu$$

```
> RES := solve({F[1],F[2],F[3],F[4],F[5]},{x,y,z,lambda,mu}) ;
```

$$RES := \left\{ z = \frac{1}{2}, y = 2, x = \frac{1}{2}, \lambda = 1, \mu = 0 \right\}$$

Controllo proprietà di minimo

```
> Hf  := Hessian(f,[x,y,z]):
```

```
Hv1 := Hessian(v1,[x,y,z]):
Hv2 := Hessian(v2,[x,y,z]):
Hf, Hv1, Hv2 ;
```

$$\begin{bmatrix} 0 & 0 & 0 \\ 0 & 0 & 0 \\ 0 & 0 & 2 \end{bmatrix}, \begin{bmatrix} 0 & 0 & 0 \\ 0 & 0 & 0 \\ 0 & 0 & 0 \end{bmatrix}, \begin{bmatrix} 0 & 0 & 0 \\ 0 & 0 & 0 \\ 0 & 0 & -4 \end{bmatrix}$$

```
> JH := Jacobian([v1,v2],[x,y,z]) ;
  NH := NullSpace(JH) ;
```

$$JH := \begin{bmatrix} 1 & 1 & 1 \\ 1 & 0 & -4\,z \end{bmatrix}$$

$$NH := \left\{ \begin{bmatrix} 4\,z \\ -4\,z - 1 \\ 1 \end{bmatrix} \right\}$$

Controllo minimo/massimo locale primo punto

```
> lambda1 := subs(RES,lambda);
  mu1     := subs(RES,mu);
```

$$\lambda 1 := 1$$

$$\mu 1 := 0$$

```
> Hf1 := simplify(Hf - lambda1. Hv1 - mu1. Hv2) ;
```

$$Hf1 := \begin{bmatrix} 0 & 0 & 0 \\ 0 & 0 & 0 \\ 0 & 0 & 2 \end{bmatrix}$$

Cerco nello spazio dei vincoli:

```
> Z1 := subs(RES,op(NH)) ;
```

$$Z1 := \begin{bmatrix} 2 \\ -3 \\ 1 \end{bmatrix}$$

E' positivo per ogni alpha > 0, quindi è un minimo locale

```
> simplify(Transpose(alpha.Z1).Hf1.(alpha.Z1)) ;
```

$$2\,\alpha^2$$

```
> subs(RES,f);
```

$$\frac{11}{4}$$

```
> evalf(%);
```

$$2.750000000$$

```
>
```